﻿package  {
	
	import flash.events.*;
	public class DeskribapenJolasa extends Jokua{
		
		private var irudiZuzena:IrudiSortzailea;
		private var irudiOkerra:IrudiSortzailea;
		private var esaldia:Esaldia;
		private var jaurtiketa:int=0;

		public function DeskribapenJolasa() {
			izena="deskribapenJolasa";
			jolastu();			
		}
		
		override public function fondoaGehitu(){
			fondoa = new Irudia("irudiak/fondoak/zura.png");
			addChildAt(fondoa,0);
		}
		
		public function jolastu(){
			jaurtiketa++;
			
			if (jaurtiketa > 10){ 
				emaitzaErakutsi(jaurtiketa,zuzenak);
				
			}else{
				irudiakHasieratu();
				irudiakKokatu();
				esaldiaPantailaratu();
				
			}
		}
		
		private function irudiakHasieratu(){
			var irudiIzenak:HitzZerrenda = HitzZerrenda.getInstantzia();
			irudiIzenak.nahastu();
			
			irudiZuzena = new IrudiSortzailea("zuzena",irudiIzenak);
			irudiZuzena.addEventListener(MouseEvent.CLICK,aukeratu);
			irudiZuzena.pantailaratu();
			irudiZuzena.x=100;
			irudiZuzena.y=50;
			
						
			irudiOkerra = new IrudiSortzailea("okerra",irudiIzenak);
			irudiOkerra.posizioakNahastu();
			irudiOkerra.pantailaratu();
			irudiOkerra.x=600;
			irudiOkerra.y=50;
			irudiOkerra.addEventListener(MouseEvent.CLICK,aukeratu);
		}
		
		private function irudiakKokatu(){
					
			var rnd :int = Math.random() * 2;
			
			if ( rnd == 1){
				irudiZuzena.x=600;
				irudiOkerra.x=100;
				
			}
			addChild(irudiZuzena);
			addChild(irudiOkerra);
			
		}
		
		private function esaldiaPantailaratu(){
			
			esaldia= new Esaldia(irudiZuzena.getEsaldia(),30);
			esaldia.x=10;
			esaldia.y=400;
			addChild(esaldia);
		}
		
		private function aukeratu(e:MouseEvent)
		{
					
			if (e.target.name == "zuzena" )
			{
				channel = ondo.play();
				zuzenak++;
				
			}else{
				
				channel = txarto.play();
			}
			super.elementuakEzabatu(this,3);
			jolastu();	
			
		}//aukeratu
		
		

	}
	
}
